

Fizică experimentală

Laborator 11

Victor E. Ambruș

*Facultatea de fizică, Universitatea de Vest din Timișoara,
Bd. Vasile Pârvan nr. 4, Timișoara, RO 300223, România*

9 ianuarie 2018

Rezumat

Pe parcursul acestui laborator, vom aborda două teme: derivarea numerică (la stânga, la dreapta și centrat) și integrarea numerică (metoda trapezelor, metoda dreptunghiurilor și metoda Simpson).

1 Introducere

2 Derivarea numerică

Prin derivarea numerică se înțelege aproximarea derivatei unei funcții $f(x)$ folosind valori ale acesteia. Inițializăm un număr de `valn + 1 = 11` puncte echidistant distribuite între $x = 0$ și $x = \pi$ cu valorile lui $f(x) = \sin x$:

```
valn:10$ xmin:0$ xmax:%pi$ hx:(xmax - xmin) / valn$  
valx:[]$ valf:[]$  
for i : 1 thru valn + 1 do (  
  valx : append(valx, [xmin + (i - 1) * hx]),  
  valf : append(valf, [sin(valx[i])])  
)$
```

În cele ce urmează, vom discuta calcularea derivatei la stânga, la dreapta și centrat.

2.1 Derivata calculată la dreapta

Pornind de la definiția derivatei la dreapta:

$$f'(x) = \lim_{\delta x \rightarrow 0} \frac{f(x + \delta x) - f(x)}{\delta x}, \quad (1)$$

se obține următoarea formulă pentru valoarea derivatei lui f în punctul x_k :

$$f'(x_k) \simeq \frac{f_{k+1} - f_k}{x_{k+1} - x_k}. \quad (2)$$

Formula de mai sus nu poate fi folosită în capătul din dreapta, când $k = \text{valn} + 1$. Implementăm formula după cum urmează:

```
derdr(valx, valf) := block([res],
  res : [],
  for i : 1 thru length(valx) - 1 do (
    res: append(res,
      [(valf[i + 1] - valf[i]) / (valx[i + 1] - valx[i])])
  ),
  append(res, [0])
)$
```

Problema 1. Să se compare grafic derivata lui f obținută numeric cu curba analitică.

2.2 Derivata calculată la stânga

Pornind de la definiția derivatei la stânga:

$$f'(x) = \lim_{\delta x \rightarrow 0} \frac{f(x) - f(x - \delta x)}{\delta x}, \quad (3)$$

se obține următoarea formulă pentru valoarea derivatei lui f în punctul x_k :

$$f'(x_k) \simeq \frac{f_k - f_{k-1}}{x_k - x_{k-1}}. \quad (4)$$

Formula de mai sus nu poate fi folosită în capătul din dreapta, când $k = 1$. Implementăm formula după cum urmează:

```
derst(valx, valf) := block([res],
  res : [0],
  for i : 2 thru length(valx) do (
    res: append(res,
      [(valf[i] - valf[i - 1]) / (valx[i] - valx[i - 1])])
  ),
  res
)$
```

Problema 2. Să se stocheze în variabila **deltast** diferența dintre valoarea derivatei lui f obținută numeric și valoarea analitică corespunzătoare.




Figura 1: Derivata funcției $f(x) = \sin x$ calculată analitic ($f'(x) = \cos x$), la dreapta, la stânga, respectiv centrat.

2.3 Derivata calculată centrat

Pornind de la definiția derivatei centrate:

$$f'(x) = \lim_{\delta x \rightarrow 0} \frac{f(x + \delta x) - f(x - \delta x)}{2\delta x}, \quad (5)$$

se obține următoarea formulă pentru valoarea derivatei lui f în punctul x_k :

$$f'(x_k) \simeq \frac{f_{k+1} - f_{k-1}}{x_{k+1} - x_{k-1}}. \quad (6)$$

Formula de mai sus nu poate fi folosită la capetele domeniului, când $k = 1$ sau $k = \text{valn} + 1$. Implementăm formula după cum urmează:

```
dermij(valx, valf) := block([res],
  res : [0],
  for i : 2 thru length(valx) - 1 do (
    res: append(res,
      [(valf[i + 1] - valf[i - 1]) / (valx[i + 1] - valx[i - 1])])
  ),
  append(res, [0])
)$
```

Problema 3. Să se compare diferențele dintre valoarea derivatei lui f obținută numeric și valoarea analitică pentru cele trei metode de derivare și să se reprezinte grafic cele trei derivate numerice împreună cu curba analitică. [Vezi Fig. 1]

Problema 4. Se observă că derivata calculată centrat se suprapune considerabil mai bine peste formula analitică comparat cu derivatele calculate la dreapta și la stânga. De ce?

3 Integrarea numerică

De data aceasta, vom folosi ca funcție $f(x) = \sin^2 x$:

```
valn:10$ xmin:0$ xmax:1$ hx:(xmax - xmin) / valn$
valx:[]$ valf:[]$
for i : 1 thru valn + 1 do (
  valx : append(valx, [xmin + (i - 1) * hx]),
  valf : append(valf, [sin(valx[i])^2])
)$
valan:(xmax - xmin) / 2 - (sin(2 * xmax) - sin(2 * xmin)) / 4$
```

Problema 4. Să se găsească folosind **Maxima** valoarea analitică a integralei lui $f(x)$ între punctele $x = 0$ și $x = 1$. [R: $\int_0^1 f(x)dx = (2 - \sin 2)/4$]

3.1 Metoda trapezelor

Metoda constă în aproximarea curbei $f(x)$ printr-o serie de linii drepte care unesc succesiv punctele (x_k, f_k) . Valoarea integralei este aproximată prin suma ariei trapezelor rezultate:

$$\int_a^b f(x) dx = \frac{1}{2} \sum_{k=1}^n [f(x_{k+1}) + f(x_k)](x_{k+1} - x_k). \quad (7)$$

Codul **Maxima** care implementează metoda trapezelor este:

```
inttrap(valx, valf) := block([val],
  val : 0,
  for i : 1 thru length(valx) - 1 do (
    val : val + (valf[i + 1] + valf[i]) / 2
      * (valx[i + 1] - valx[i])
  ),
  val
)$
```

Problema 5. Să se compare rezultatul numeric cu cel analitic. [0, 2734 vs. 0, 2727]

Problema 6. Să se refacă comparația folosind 20 de puncte. [0, 2729 vs. 0, 2727]

3.2 Metoda dreptunghiurilor

Dacă numărul de intervale **valn** este par, se poate folosi metoda dreptunghiurilor, care aproximează graficul funcției între x_{k-1} și x_{k+1} printr-o linie dreaptă având ordonata egală cu f_k :

$$\int_a^b f(x) dx = \frac{1}{2} \sum_{k=1}^{n/2} f_{2k}(x_{2k+1} - x_{2k-1}). \quad (8)$$




Figura 2: Dependența lui ϵ (10) de numărul de puncte în cazul metodelor trapezelor, dreptunghiurilor și Simpson compozită.

Codul Maxima care implementează metoda dreptunghiurilor este:

```
intdrep(valx, valf) := block([val],
  val : 0,
  for i : 2 thru length(valx) - 1 step 2 do (
    val : val + valf[i] * (valx[i + 1] - valx[i - 1])
  ),
  val
)$
```

Problema 7. Să se calculeze rezultatul numeric obținut folosind metoda dreptunghiurilor pentru 10 puncte, respectiv 20 de puncte și să se compare cu rezultatul analitic. [0,2712 vs. 0,2723 vs. 0,2727]

3.3 Metoda Simpson compozită

În cazul în care punctele sunt echidistante, având distanța $x_{k+1} - x_k = h$ între ele, se poate utiliza metoda Simpson compozită, conform căreia integrala pe domeniul $[x_{k-1}, x_{k+1}]$ se aproximează prin:

$$\int_{x_{k-1}}^{x_{k+1}} dx f(x) \simeq \frac{h}{3} [f_{k+1} + 4f_k + f_{k-1}]. \quad (9)$$

Ca și în cazul metodei dreptunghiurilor, numărul de intervale trebuie să fie par. Codul Maxima se poate regăsi mai jos:

```
intsimp(valx, valf) := block([val],
  val : 0,
  for i : 2 thru length(valx) - 1 step 2 do (
```

```

        val : val + (valf[i + 1] + 4 * valf[i] + valf[i - 1])
                * (valx[i + 1] - valx[i - 1]) / 6
    ),
    val
)$

```

Problema 7. Să se compare rezultatele obținute cu 10, respectiv 20 de puncte cu valoarea analitică. [R: 0,27267361 vs. 0,27267552 vs. 0,27267564]

Problema 8. Se observă că, la același număr de puncte, metoda Simpson dă rezultatele cele mai bune. Pornind de la definiția erorii relative:

$$\epsilon = (\text{numeric}/\text{analitic}) - 1, \quad (10)$$

să se investigheze dependența acesteia față de numărul de puncte în scară dublu logaritmică pentru evidențierea ordinului metodelor trapezelor, dreptunghiurilor și Simpson compizită. [R: Vezi Fig. 2]