
Rezumat

Pe parcursul acestui laborator, vom aborda două teme: derivarea nu-
merică (la stânga, la dreapta s, i centrat) s, i integrarea numerică (metoda
trapezelor, metoda dreptunghiurilor s, i metoda Simpson).

1 Introducere

2 Derivarea numerică

Prin derivarea numerică se ı̂nt,elege aproximarea derivatei unei funct, ii f(x) fo-
losind valori ale acesteia. Init, ializăm un număr de valn + 1 = 11 puncte echi-
distant distribuite ı̂ntre x = 0 s, i x = π cu valorile lui f(x) = sinx:

valn:10$ xmin:0$ xmax:%pi$ hx:(xmax - xmin) / valn$

valx:[]$ valf:[]$

for i : 1 thru valn + 1 do (

valx : append(valx, [xmin + (i - 1) * hx]),

valf : append(valf, [sin(valx[i])])

)$

În cele ce urmează, vom discuta calcularea derivatei la stânga, la dreapta s, i
centrat.

2.1 Derivata calculată la dreapta

Pornind de la definit, ia derivatei la dreapta:

f ′(x) = lim
δx→0

f(x+ δx)− f(x)

δx
, (1)

se obt, ine următoarea formulă pentru valoarea derivatei lui f ı̂n punctul xk:

f ′(xk) ' fk+1 − fk
xk+1 − xk

. (2)

Formula de mai sus nu poate fi folosită ı̂n capătul din dreapta, când k = valn

+ 1. Implementăm formula după cum urmează:

derdr(valx, valf):= block([res],

res : [],

for i : 1 thru length(valx) - 1 do (

res: append(res,

[(valf[i + 1] - valf[i]) / (valx[i + 1] - valx[i])])

),

append(res, [0])

)$

Problema 1. Să se compare grafic derivata lui f obt, inută numeric cu curba
analitică.

2.2 Derivata calculată la stânga

Pornind de la definit, ia derivatei la stânga:

f ′(x) = lim
δx→0

f(x)− f(x− δx)

δx
, (3)

se obt, ine următoarea formulă pentru valoarea derivatei lui f ı̂n punctul xk:

f ′(xk) ' fk − fk−1
xk − xk−1

. (4)

Formula de mai sus nu poate fi folosită ı̂n capătul din dreapta, când k = 1.
Implementăm formula după cum urmează:

derst(valx, valf):= block([res],

res : [0],

for i : 2 thru length(valx) do (

res: append(res,

[(valf[i] - valf[i - 1]) / (valx[i] - valx[i - 1])])

),

res

)$

Problema 2. Să se stocheze ı̂n variabila deltast diferent,a dintre valoarea
derivatei lui f obt, inută numeric s, i valoarea analitică corespunzătoare.

Figura 1: Derivata funct, iei f(x) = sinx calculată analitic (f ′(x) = cosx), la
dreapta, la stânga, respectiv centrat.

2.3 Derivata calculată centrat

Pornind de la definit, ia derivatei centrate:

f ′(x) = lim
δx→0

f(x+ δx)− f(x− δx)

2δx
, (5)

se obt, ine următoarea formulă pentru valoarea derivatei lui f ı̂n punctul xk:

f ′(xk) ' fk+1 − fk−1
xk+1 − xk−1

. (6)

Formula de mai sus nu poate fi folosită la capetele domeniului, când k = 1 sau
k = valn + 1. Implementăm formula după cum urmează:

dermij(valx, valf):= block([res],

res : [0],

for i : 2 thru length(valx) - 1 do (

res: append(res,

[(valf[i + 1] - valf[i - 1]) / (valx[i + 1] - valx[i - 1])])

),

append(res, [0])

)$

Problema 3. Să se compare diferent,ele dintre valoarea derivatei lui f
obt, inută numeric s, i valoarea analitică pentru celei trei metode de derivare s, i
să se reprezinte grafic cele trei derivate numerice ı̂mpreună cu curba analitică.
[Vezi Fig. 1]

Problema 4. Se observă că derivata calculată centrat se suprapune con-
siderabil mai bine peste formula analitică comparat cu derivatele calculate la
dreapta s, i la stânga. De ce?

3 Integrarea numerică

De data aceasta, vom folosi ca funct, ie f(x) = sin2 x:

valn:10$ xmin:0$ xmax:1$ hx:(xmax - xmin) / valn$

valx:[]$ valf:[]$

for i : 1 thru valn + 1 do (

valx : append(valx, [xmin + (i - 1) * hx]),

valf : append(valf, [sin(valx[i])^2])

)$

valan:(xmax - xmin) / 2 - (sin(2 * xmax) - sin(2 * xmin)) / 4$

Problema 4. Să se găsească folosind Maxima valoarea analitică a integralei

lui f(x) ı̂ntre punctele x = 0 s, i x = 1. [R:
∫ 1

0
f(x)dx = (2− sin 2)/4]

3.1 Metoda trapezelor

Metoda constă ı̂n aproximarea curbei f(x) printr-o serie de linii drepte care
unesc succesiv punctele (xk, fk). Valoarea integralei este aproximată prin suma
ariei trapezelor rezultate:∫ b

a

f(x) dx =
1

2

n∑
k=1

[f(xk+1) + f(xk)](xk+1 − xk). (7)

Codul Maxima care implementează metoda trapezelor este:

inttrap(valx, valf) := block([val],

val : 0,

for i : 1 thru length(valx) - 1 do (

val : val + (valf[i + 1] + valf[i]) / 2

* (valx[i + 1] - valx[i])

),

val

)$

Problema 5. Să se compare rezultatul numeric cu cel analitic. [0, 2734 vs.
0, 2727]

Problema 6. Să se refacă comparat, ia folosind 20 de puncte. [0, 2729 vs.
0, 2727]

3.2 Metoda dreptunghiurilor

Dacă numărul de intervale valn este par, se poate folosi metoda dreptunghiuri-
lor, care aproximează graficul funct, iei ı̂ntre xk−1 s, i xk+1 printr-o linie dreaptă
având ordonata egală cu fk:∫ b

a

f(x) dx =
1

2

n/2∑
k=1

f2k(x2k+1 − x2k−1). (8)

Figura 2: Dependent,a lui ε (10) de numărul de puncte ı̂n cazul metodelor tra-
pezelor, dreptunghiurilor s, i Simpson compozită.

Codul Maxima care implementează metoda dreptunghiurilor este:

intdrep(valx, valf) := block([val],

val : 0,

for i : 2 thru length(valx) - 1 step 2 do (

val : val + valf[i] * (valx[i + 1] - valx[i - 1])

),

val

)$

Problema 7. Să se calculeze rezultatul numeric obt, inut folosind metoda
dreptunghiurilor pentru 10 puncte, respectiv 20 de puncte s, i să se compare cu
rezultatul analitic. [0, 2712 vs. 0, 2723 vs. 0, 2727]

3.3 Metoda Simpson compozită

În cazul ı̂n care punctele sunt echidistante, având distant,a xk+1 − xk = h ı̂ntre
ele, se poate utiliza metoda Simpson compozită, conform căreia integrala pe
domeniul [xk−1, xk+1] se aproximează prin:∫ xk+1

xk−1

dx f(x) ' h

3
[fk+1 + 4fk + fk−1]. (9)

Ca s, i ı̂n cazul metodei dreptunghiurilor, numărul de intervale trebuie să fie par.
Codul Maxima se poate regăsi mai jos:

intsimp(valx, valf) := block([val],

val : 0,

for i : 2 thru length(valx) - 1 step 2 do (

val : val + (valf[i + 1] + 4 * valf[i] + valf[i - 1])

* (valx[i + 1] - valx[i - 1]) / 6

),

val

)$

Problema 7. Să se compare rezultatele obt, inute cu 10, respectiv 20 de
puncte cu valoarea analitică. [R: 0, 27267361 vs. 0, 27267552 vs. 0, 27267564]

Problema 8. Se observă că, la acelas, i număr de puncte, metoda Simpson
dă rezultatele cele mai bune. Pornind de la definit, ia erorii relative:

ε = (numeric/analitic)− 1, (10)

să se investigheze dependent,a acesteia fat, ă de numărul de puncte ı̂n scară dublu
logaritmică pentru evident, ierea ordinului metodelor trapezelor, dreptunghiurilor
s, i Simpson compizită. [R: Vezi Fig. 2]

