
Rezumat

Pe parcursul acestui laborator vor fi trecute ı̂n revistă următoarele
tehnici de interpolare s, i extrapolare: regresia liniară, polinomul de inter-
polare Lagrange s, i interpolarea spline.

1 Datele experimentale

Să considerăm setul de date din tab. 1, pe care le putem introduce ı̂n Maxima

cu următoarele comenzi:

valx:[0.0, 0.1, 0.2, 0.3, 0.4,0.5, 0.6, 0.7, 0.8, 0.9, 1.0]$

valf:[0.03,0.098,0.197,0.305,0.4,0.492,0.61,0.689,0.81,0.899,1.01]$

Putem reprezenta aceste date cu comanda:

plot2d([discrete, valx, valf],

Figura 1: (a) Datele experimentale. (b) Dreapta de regresie liniară.

x y

0.0 0.03
0.1 0.098
0.2 0.197
0.3 0.305
0.4 0.4
0.5 0.492
0.6 0.61
0.7 0.689
0.8 0.81
0.9 0.899
1.0 1.01

Tabela 1: Datele experimentale.

[legend, "Experiment"],

[style, points],

[xlabel, "x"],

[ylabel, "y"]);

Rezultatul poate fi vizualizat ı̂n fig. 1.

2 Regresia liniară

Regresia liniară este una dintre cele mai uzitate metode de interpolare s, i ex-
trapolare a datelor experimentale. Regresia liniară are la bază metoda celor
mai mici pătrate, putând fi extinsă cu us,urint, ă la cazuri mai complexe (regresie
pătratică, etc).

Regresia liniară pleacă de la presupunerea că dependent,a y(x) este de tip
liniar:

y(x) = ax+ b. (1)

În acest caz, metoda celor mai mici pătrate dă următoarele valori optime pentru
a s, i b:

a =
xy − x y
x2 − (x)2

, b =
x2 y − x xy
x2 − (x)2

. (2)

Pentru aflarea coeficient, ilor a s, i b avem nevoie de calcularea unor medii. Cal-
culul mediilor poate fi făcut utilizând comanda mean din pachetul descriptive.
Începem prin ı̂ncărcarea acestui pachet:

load (descriptive);

Comanda mean ia ca s, i parametru un s, ir sau o matrice s, i returnează media
acestuia. Mediile lui valx s, i valy le vom stoca ı̂n variabilele mx s, i my:

2



mx:mean(valx); my:mean(valy);

Putem calcula celelalte medii din ec. (2) t, inând cont că s, irurile se ı̂nmult,esc
element cu element utilizând operat, ia *:

mxy:mean(valx*valy); mxx:mean(valx*valx);

Problema 1. Să se definească variabilele rega s, i regb având valorile lui a
s, i b din ec. (2). [R: a ' 0, 993, b ' 0, 007]

Problema 2. Să se reprezinte grafic dreapta de regresie liniară ı̂mpreună
cu punctele experimentale.

Solut, ie:

plot2d([[discrete, valx, valy],

rega*x+regb], [x,0,1],

[legend, "Experiment", "Regresie liniara"],

[style, points, lines],

[xlabel, "x"],

[ylabel, "y"]);

Rezultatul poate fi consultat ı̂n fig. 1(b).
Metoda celor mai mici pătrate poate fi implementată s, i analitic. Pentru

aceasta, definim funct, ia S({x}, {y}):

S({x}, {y}) =

N∑
i=1

[yi − y(x)]2, (3)

aceasta reprezentând suma pătratelor distant,elor dintre valorile yi măsurate
experimental s, i cele obt, inute prin regresia liniară (1). Metoda celor mai mici
pătrate caută valorile a s, i b pentru care S({x}, {y}) are valoare minimă:

∂S

∂a
= 0,

∂S

∂b
= 0. (4)

În Maxima, putem construi valS ca fiind valoarea funct, iei S (3):

valS : sum((valy[i] - (a * valx[i] + b))^2, i, 1, length(valx));

Mai sus am folosit funct, ia sum, care ı̂nsumează valorile expresiei din primul
argument obt, inute pentru toate valorile contorului de pe pozit, ia a doua, care
sunt cuprinse ı̂ntre limitele date pe pozit, iile 3 s, i 4. Se vede că valS cont, ine
variabilele a s, i b la nivel simbolic.

Problema 3. Să se găsească valorile lui a s, i b rezolvând sistemul de ecuat, ii
dS/da = 0, dS/db = 0. [a ' 0.993, b ' 0.007]

3 Polinomul de interpolare Lagrange

Orice funct, ie care este determinată ca urmare a unui proces de măsurare are
un număr finit N de perechi (xk, yk), unde k = 1, 2, . . . N , denumite puncte

Figura 2: (a) Polinomul de interpolare Lagrange. (b) Interpolarea folosind
funct, ii spline de ordinul 3 (cubice).

experimentale. Polinomul de interpolare Lagrange L(x) se scrie:

L(x) =

N∑
k=1

ykuk(x), (5)

unde uk(x) sunt polinoame de ordinul N − 1, definite prin:

uk(x) =
∏
j 6=k

x− xj
xk − xj

. (6)

Se observă că uk(xj) = 0 pentru toate valorile lui j diferite de k (când j 6= k),
ı̂n timp ce uk(xk) = 1. O proprietate specială a polinomului de interpolare
Lagrange este că valoarea acestuia ı̂n punctele x = xk este cea măsurată expe-
rimental, adică L(xk) = yk.

3.1 Folosind pachetul interpol

Maxima are deja implementată funct, ia care dă polinomul Lagrange pentru un
set de puncte. În primul rând, trebuie ı̂ncărcat pachetul interpol, unde e
definită această funct, ie:

load(interpol)$

Funct, ia care ne interesează se numes,te lagrange s, i ia ca argument un singur
parametru, structurat sub forma unei liste de perechi de valori (xk, yk). Creăm
această listă după cum urmează:

puncte : block([pct,k],

pct:[],

for k : 1 thru length(valx) do (

pct : append(pct, [[valx[k], valy[k]]])

),

pct

);

Mai sus am introdus structura for. Polinomul Lagrange se obt, ine folosind:

lag : lagrange(puncte)$

Problema 4. Să se reprezinte grafic polinomul de interpolare Lagrange, ı̂mpreună
cu punctele experimentale. [Vezi fig. 2(a)]

3.2 Folosind definit, ia

În Maxima, aceste polinoamele uk(x) (6) se pot defini folosind funct, ia:

func_uk(valx,k) := block([uk, j],

uk : 1,

for j : 1 thru length(valx) do (

if notequal(j, k) then uk : uk * (x - valx[j]) / (valx[k] - valx[j])

),

return (uk)

)$

Polinomul de interpolare Lagrange (5) se scrie:

func_lag(valx, valy) := block([k, lag],

lag : 0,

for k : 1 thru length(valx) do (

lag : lag + valy[k] * func_uk(valx, k)

),

lag

)$

Problema 5. Să se verifice dacă rezultatul obt, inut mai sus coincide cu cel
returnat de funct, ia lagrange.

4 Interpolarea cu funct, ii spline cubice

Pentru funct, ii nemonotone, interpolarea Lagrange poate duce la abateri im-
portante ale functiei L(x) fat, ă de f(x). În astfel de cazuri, o aproximare mai
bună se obt, ine prin interpolarea cu ajutorul funct, iilor spline. Funct, iile spline de
ordinul m reprezintă polinoame de ordinul m astfel construite ı̂ncât derivatele
acestora până la ordinul m− 1 să fie continue ı̂n punctele xk (noduri).

Pentru construirea interpolării spline aferente unui set de N = n+ 1 puncte
experimentale, se utilizează un număr de n funct, ii Sk (k = 1, 2, . . . n), fiecare
fiind definită pe intervalul [xk, xk+1] corespunzător.

În cazul funct, iilor spline cubice (m = 3), se foloses,te notat, ia:

Sk(x) = pk + qk(x− xk) + rk(x− xk)2 + sk(x− xk)3, x ∈ [xk, xk+1]. (7)

Pentru determinarea celor 4n coeficient, i constant, i pk, qk, rk s, i sk avem nevoie
de 4n relat, ii independente. În primul rând, funct, iile Sk(x) trebuie să treacă
prin punctele care mărginesc intervalul pe care ı̂l interpolează:

Sk(xk) = yk, Sk(xk+1) = yk+1. (8)

Rezultă 2n ecuat, ii. Alte 2n− 2 ecuat, ii se pot obt, ine impunând ca derivatele de
ordinul 1 s, i 2 să fie continue ı̂n punctele xk (2 ≤ k ≤ n):

S′k(xk+1) =S′k+1(xk+1), continuitatea pantei

S′′k (xk+1) =S′′k+1(xk+1), continuitatea pantei, (9)

Adunând constrângerile enumerate mai sus, rezultă că rămânem cu două grade
de libertate, care sunt fixate de condit, iile:

S′′1 (x1) =0 ⇒ r1 =0,

S′′n(xn+1) =0 ⇒ sn =− rn
3hn

, (10)

unde s-a introdus notat, ia:

hk = xk+1 − xk, 1 ≤ k ≤ n. (11)

4.1 Folosind pachetul interpol

Folosind puncte, interpolul ne dă spline-ul folosind comanda:

spl : cspline(puncte)

Problema 6. Să se reprezinte grafic funct, ia de interpolare spline, ı̂mpreună cu
punctele experimentale. [Vezi fig. 2(b)]

4.2 Folosind gnuplot

În gnuplot, reprezentarea unui set de date folosind interpolarea cu funct, ii spline
cubice se face folosind opt, iunea smooth mcsplines:

plot "date.dat" u 1:2 w p ps 2 pt 7, "" w l lw 3 smooth mcsplines

4.3 Folosind recurent,a

Folosind Sk(xk) = yk, rezultă imediat:

pk = yk, 1 ≤ k ≤ n. (12)

Mai departe, condit, ia S′′k (xk+1) = S′′k+1(xk+1) permite eliminarea lui sk:

sk =
rk+1 − rk

3hk
, 1 ≤ k ≤ n− 1, (13)

unde sn = −rn/3hn.
Din condit, ia Sk(xk+1) = yk+1 rezultă:

qk = bk +
rk+1 + 2rk

3
hk, 1 ≤ k ≤ n− 1, (14)

unde s-a introdus notat, ia:

bk =
yk+1 − yk

hk
. (15)

Când k = n, avem:

qn = bn −
2

3
rnhn. (16)

În fine, condit, ia S′k(xk+1) = S′k+1(xk+1) impune:

hk+1rk+2 + 2(hk + hk+1)rk+1 + hkrk = 3(bk+1 − bk), 1 ≤ k ≤ n− 2, (17)

ı̂mpreună cu condit, iile:

r1 = 0, 2rn(hn + hn−1) + hn−1rn−1 = 3(bn − bn−1). (18)

Ultima ecuat, ie e compatibilă cu ec. (17) s, i (14) dacă se consideră rn+1 = 0
(această alegere este formală, ı̂ntrucât funct, iile spline nu sunt definite pentru
k = n + 1, ı̂nsă ne ajută la scrierea recurent,elor). Drept urmare, coeficient, ii
rk se pot găsi rezolvând ecuat, ia (17) cu 1 ≤ k ≤ n − 1, ı̂mpreună cu condit, iile
r1 = rn+1 = 0.

Dezavantajul ec. (17) este că condit, iile la limită sunt ı̂n capetele opuse k = 1
s, i k = n + 1. Deoarece rk satisface astfel de condit, ii la limită, se poate arăta
că ec. (17) este echivalentă cu o recurent, ă cu doi termeni, care poate fi scrisă
astfel:

hkrk+1 + αkrk = βk, 2 ≤ k ≤ n− 1. (19)

Pentru k = 2 avem:
h2r3 + α2r2 = β2. (20)

Comparând cu ec. (17) pentru k = 1, rezultă condit, iile init, iale pentru αk s, i βk:

α2 = 2(h1 + h2), β2 = 3(b2 − b1). (21)

Scriind ec. (19) pentru k → k + 1 rezultă:

hk+1rk+2 + αk+1rk+1 = βk+1. (22)

Înmult, ind ec. (19) cu [2(hk + hk+1) − αk+1]/hk s, i adunând cu relat, ia de mai
sus, se obt, ine:

hk+1rk+2 + 2(hk + hk+1)rk+1 +
αk

hk
[2(hk + hk+1)− αk+1]rk

= βk+1 +
βk
hk

[2(hk+1 + hk)− αk+1]. (23)

Prin comparat, ie cu ec. (17), deducem relat, iile de recurent, ă pentru αk s, i βk:

αk+1 = 2(hk +hk+1)− h2k
αk
, βk+1 = 3(bk+1− bk)− hk

αk
βk, 2 ≤ k ≤ n−1.

(24)
După determinarea s, irurilor αk s, i βk, s, irul rk se poate determina pornind de la
k = n (cunoscând rn+1 = 0) folosind ec. (19). Având s, irul rk, se pot construi
sk s, i qk cu ajutorul ec. (13) s, i (14). În fine, ec. (12) implică pk = yk. Mai jos
prezentăm pas, ii unui algoritm Maxima pentru construct, ia interpolării cu funct, ii
spline cubice.

1. Se defines,te setul de numere hk = xk+1 − xk (1 ≤ k ≤ n):

hk : block([hk],

hk : [],

for k : 1 thru length(valx) - 1 do (

hk : append(hk, [valx[k + 1] - valx[k]])),

hk)$

2. Se defines,te setul de numere bk = (yk+1 − yk)/hk:

bk : block([bk],

bk : [],

for k : 1 thru length(valx) - 1 do (

bk : append(bk, [(valy[k + 1] - valy[k]) / hk[k]])),

bk)$

3. Mai departe, avem nevoie de s, irul auxiliar αk (2 ≤ k ≤ n), definit prin
(24):

alphak : block([auxalpha,alphak],

auxalpha : 2 * (hk[2] + hk[1]),

alphak : [0, auxalpha],

for k : 3 thru length(hk) do (

auxalpha : 2 * (hk[k] + hk[k-1]) - hk[k-1]^2 / auxalpha,

alphak : append(alphak, [auxalpha])),

alphak)$

4. Al doilea s, ir auxiliar, βk (2 ≤ k ≤ n), este definit tot prin (24):

betak : block([auxbeta,betak],

betak : [auxbeta : 0],

for k : 2 thru length(hk) do (

auxbeta : 3 * (bk[k] - bk[k - 1]) - auxbeta *

(2 * (hk[k] + hk[k-1]) - alphak[k]) / hk[k - 1],

betak : append(betak, [auxbeta])),

betak)$

8



5. În sfârs, it, putem determina coeficient, ii rk (1 ≤ k ≤ n) rezolvând ec. (19)
pornind de la k = n, descendent către k = 1:

rk : block([auxr, rk],

rk : [auxr : 0],

for k : length(hk) thru 2 step -1 do (

auxr : (betak[k] - auxr * hk[k]) / alphak[k],

rk : append([auxr], rk)),

append([0], rk))$

6. Coeficient, ii pk se determină din condit, ia (12):

pk : valy$

7. Coeficient, ii sk se determină din relat, ia (13):

sk : block([sk],

sk : [],

for k : 1 thru length(hk) do (

sk : append(sk, [(rk[k + 1] - rk[k]) / 3 / hk[k]])

),

sk)$

8. Coeficient, ii qk se determină din relat, ia (14):

qk : block([qk],

qk : [],

for k : 1 thru length(hk) do (

qk : append(qk, [bk[k] - (2 * rk[k] + rk[k + 1]) * hk[k] / 3])

),

qk)$

După urmarea pas, ilor de mai sus, se poate scrie o funct, ie care evaluează funct, ia
spline ı̂n orice punct:

spline_eval(x) := block([startk, dx],

for k : 2 thru length(valx) do (

startk : k - 1,

if (x <= valx[k]) then return (startk)

),

dx : x - valx[startk],

pk[startk] + qk[startk] * dx + rk[startk] * dx^2 + sk[startk] * dx^3

)$

Mai departe, generăm o listă cu npuncte puncte ı̂n care evaluăm funct, ia spline:

npuncte:100$

puncte_spline : (block[auxpuncte, auxx],

auxpuncte: [],

for k : 1 thru (npuncte + 1) do (

auxx : valx[1] + (valx[length(valx)] - valx[1]) * (k - 1) / npuncte,

auxpuncte : append(auxpuncte, [[auxx, spline_eval(auxx)]])),

auxpuncte

)$

În fine, afis, ăm capodopera:

plot2d([[discrete, valx,valy],[discrete, puncte_spline]],

[x, valx[1], valx[length(valx)]],

[legend, "Exp", "Spline"],

[style, points, lines]);

Rezultatul poate fi vizualizat ı̂n fig. 2(b).

