
Rezumat

Pe parcursul acestui laborator, ne vom familiariza cu comenzile pentru
rezolvarea ecuat, iilor algebrice s, i a celor diferent, iale, atât prin metode
analitice, cât s, i numeric. Vom considera mis,carea circulară s, i traiectoria
proiectilului ı̂n mediu rezistiv s, i nerezistiv.

1 Mis,carea circulară

1.1 Definirea unei funct, ii

Legea de mis,care a mis,cării circulare este:

r(t) = x(t)i + y(t)j, (1)

unde x(t) s, i y(t) sunt funct, ii armonice de t:

x(t) = R cosωt, y(t) = R sinωt. (2)

Putem defini ecuat, iile de mis,care de mai sus sub forma unor funct, ii:

xp(t) := R * cos(t * omega);

yp(t) := R * sin(t * omega);

Mai sus am definit funct, iile xp(t) s, i yp(t), care descriu dependent,a de timp
a coordonatelor x s, i y ale particulei. Pentru definirea unei funct, ii se foloses,te
operatorul := (de remarcat că variabilele se definesc folosind operatorul :).

1.2 Instruct, iunea subst

Putem afis,a traiectoria corespunzătoare ec. (2) folosind următoarea comandă:

Figura 1: Traiectoria circulară corespunzătoare ec. (2).

plot2d([parametric,

subst([R=1, omega=2*%pi], xp(t)),

subst([R=1, omega=2*%pi], yp(t)),

[t,0,1]],

[xlabel,"x/R"],[ylabel,"y/R"],

[title, "Traiectorie circulara"],

[same_xy, true]);

Să analizăm comanda de mai sus.

• Comanda plot2d generează un grafic folosind gnuplot.

• Primul argument este {[parametric, x, y, [t, tmin, tmax]]}. Gra-
ficele parametrice reprezintă curba având coordonatele (x, y) pentru valo-
rile parametrului t specificat pe pozit, ia a patra, care ia valori ı̂ntre tmin

s, i tmax.

• Coordonatele x s, i y sunt date de funct, iile xp(t) s, i yp(t), ı̂nsă aceste
funct, ii au fost definite de parametrii nespecificat, i R s, i omega. Pentru a
putea reprezenta grafic traiectoria aferentă acestor funct, ii, trebuie să dăm
valori numerice lui R s, i ω.

• Putem atribui valori numerice acestor parametri folosind comanda subst.
Comanda subst ia doi parametri: primul parametru reprezintă o listă de
atribuiri de valori de tipul [var1 = val1, var2 = val2,...], ı̂n timp
ce parametrul al doilea reprezintă expresia ı̂n care se fac substitut, iile.

• Funct, ia plot2d mai acceptă parametri care pot fi folosit, i pentru configu-
rarea descrierii axelor graficului ([xlabel, "x/R"] s, i [ylabel, "y/R"])
sau a titlului graficului ([title, "Traiectorie circulara"]).

2



y/
R

x/R

cerculet

-1

-0.5

	0

	0.5

	1

-1 -0.5 	0 	0.5 	1

y/
R

x/R

cerculet

-1

-0.5

	0

	0.5

	1

-1 -0.8 -0.6 -0.4 -0.2 	0 	0.2 	0.4 	0.6 	0.8 	1

(a) (b)

Figura 2: Figurile Lissajous care rezultă din utilizarea unor pulsat, ii diferite pe
axele x s, i y ı̂n ec. (2). (a) ωx = 6π, ωy = 4π; (b) ωx = 2π2, ωy = 2π.

• Ultimul parametru al comenzii plot2d este [same_xy, true], care im-
pune aceeas, i scală pe cele două axe (astfel ı̂ncât traiectoria se va vedea
ı̂ntr-adevăr ca un cerc).

1.3 Instructiunea block

Să considerăm acum o mis,care plană care reprezintă compunerea a două oscilat, ii
cu perioade s, i amplitudini nu neapărat egale. Mai exact, ne interesează următoarea
traiectorie:

x(t) = Rx cos(ωxt), y(t) = Ry cos(ωyt). (3)

Traiectoria rezultantă poartă numele de figură Lissajous. Putem reprezenta
această traiectorie folosind funct, iile xp(t) s, i yp(t), ı̂nsă dând valori diferite
pentru R s, i omega cu ajutorul comenzii subst.

Să presupunem că dorim să ı̂ncercăm combinat, ii multiple ale valorilor lui
Rx, Ry, ωx s, i ωy. În acest caz, este convenabilă utilizarea unei structuri de tip
block, ı̂n care să putem defini valorile acestea folosind nis,te variabile locale:

block([Rx,Ry,omegax,omegay],

Rx:1,Ry:1,omegax:6*%pi,omegay:4*%pi,

plot2d([parametric,

subst([R=Rx,omega=omegax],xp(t)),

subst([R=Ry,omega=omegay],yp(t)),

[t,0,1]],

[xlabel,"x/R"],[ylabel,"y/R"],

[title, "Compunere"],

[nticks,1000],

[same_xy, true])

);

Mai sus am definit un bloc de instruct, iuni folosind comanda block. Pe prima
pozit, ie am specificat lista de variabile locale, s, i anume Rx, Ry, omegax s, i omegay.

Figura 3: (a) Valoarea maximă a bătăii adimensionalizate rmax0 ca funct, ie de
θ pentru k = 2; (b) Valoarea unghiului θ sub care este lansat proiectilul pentru
a obt, ine bătaia maximă ca funct, ie de k.

Mai departe am utilizat comanda plot2d discutată anterior, de data aceasta
substituind R = Rx s, i ω = ωx pentru coordonata orizontală xp(t), respectiv
R = Ry s, i ω = ωy pentru coordonata verticală yp(t). Am mai specificat prin
parametrul [nticks, 1000] să fie folosite 1000 de puncte pentru generarea
traiectoriei, pentru a evita aspectul colt,uros al graficului.

Problema 1. Să se utilizeze instruct, iunile de mai sus pentru a reprezenta
figura Lissajous corespunzătoare compunerii a două mis,cări armonice de ampli-
tudini Rx = Ry = 1, având pulsat, iile ωx = 2π2 s, i ωy = 2π. Să se folosească
domeniul temporal t ∈ [0, 20]. [Vezi fig. 2(b).]

2 Aruncarea oblică

Pentru studierea mis,cării oblice, definim legea de mis,care a unui corp sub ac-
tiunea unei accelerat, ii constante ı̂n sensul negativ al axei y:

xp(t) := x0 + v0 * cos(theta) * t;

yp(t) := y0 + v0 * sin(theta) * t - g * t^2 / 2;

Problema 2. Să se găsească momentul de timp t la care proiectilul atinge
solul (y = 0).

Solut, ie:

solve(yp(t)=0,t);

Se observă că instruct, iunea de mai sus returnează două valori (una pozitivă s, i
una negativă). Pentru a salva coordonata acestui punct, putem folosi instruct, iunea:

tmax:rhs(%[2]);

Comanda rhs returnează membrul drept al unei egalităt, i, ı̂n timp ce %[2] se
referă la elementul al doilea din s, irul returnat ı̂n urma execut, iei ultimei comenzi.

Cele două comenzi de mai sus (solve s, i rhs) pot fi combinate ı̂ntr-o singură
comandă care returnează direct valoarea lui tmax:

tmax: rhs(solve(yp(t)=0,t)[2]);

Problema 3. Să se găsească distant,a fat, ă de punctul de aruncare când
proiectilul atinge solul (bătaia).

Solut, ie: Distant,a parcursă până la atingerea solului (bătaia) se poate calcula
folosind tmax:

rmax:xp(tmax)-x0;

Problema 4. Să se găsească momentul de timp la care ı̂nălt, imea proiectilul
este maximă. Să se atribuie această valoare variabilei tymax. [R:
tymax = (sin(theta)*v0)/g]

Problema 5. Să se găsească ı̂nălt, imea maximă pe care o atinge proiectilul.
Să se atribuie această valoare variabilei ymax. [R:
ymax = y0+(sin(theta)^2*v0^2)/(2*g)]

În continuare, este convenabil să introducem o scală h0 pentru ı̂nălt, ime,
definită ca fiind ı̂nălt, imea la care energia potent, ială a proiectilului este egală cu
energia sa cinetică init, ială:

h0 =
v20
2g
. (4)

Cu ajutorul acestei scale de lungime, putem defini y0 = kh0 (k fiind ı̂nălt, imea
init, ială ı̂n unităt, i de h0). Cu ajutorul lui h0 s, i k, putem defini bătaia adimen-
sională ca raportul rmax/2h0, după cum urmează:

rmax0:radcan(subst([v0=sqrt(2*g*h0),y0=k*h0],rmax)/(2*h0));

Funct, ia radcan a fost introdusă pentru a simplifica expresia rezultantă.
Problema 6. Să se reprezinte grafic rmax0 ca funct, ie de unghiul θ, exprimat

ı̂n grade, sub care proiectilul este aruncat pentru cazul când k = 2. [Vezi
fig. 3(a).]

Solut, ie:

plot2d(subst([k=2,theta=thdeg*%pi/180], rmax0), [thdeg,0,90],

[xlabel, "{/Symbol q} (deg)"],

[ylabel, "r_{max, 0}"],

[title, "Bataia adimensionala"]);

Comanda de mai sus face ı̂n instructiunea subst o conversie a lui θ din valoarea
exprimată ı̂n grade (thdeg) ı̂ntr-o valoare exprimată ı̂n radiani (θ×π/180◦), iar
rezultatul se atribuie necunoscutei theta. Descrierea axei x este θ(deg), unde
simbolul grecesc θ se obt, ine ı̂n gnuplot folosind instruct, iunea {/Symbol q}.

Problema 7. Să se găsească valoarea lui θ pentru care rmax0 este maxim
pentru cazul când k = 2. Să se exprime rezultatul ı̂n grade.

Solut, ie: În acest caz, comanda solve nu va găsi analitic valoarea lui θ pentru
care rmax,0 este maxim. De aceea, vom folosi comanda find_root:

Figura 4: Comparat, ie ı̂ntre componentele orizontale (a) s, i verticale (b) ale vi-
tezelor ı̂n aruncarea oblică cu rezistent, ă liniară s, i fără rezistent, ă.

float(find_root(diff(subst(k=2,rmax0),theta),theta,0,%pi/2)*180/%pi);

=> 30.0

Putem reprezenta grafic valoarea lui θ (̂ın grade) la care rmax0 este maxim
ca funct, ie de k:

plot2d(float(find_root(diff(rmax0,theta),theta,0,%pi/2)*180/%pi),

[k,0.01,50],[xlabel,"k"],[ylabel,"{/Symbol q}_{max}(deg)"]);

Rezultatul acestei comenzi se poate vizualiza ı̂n fig. 3(b).

3 Aruncarea oblică ı̂n mediu cu rezistivitate li-
niară

Să considerăm problema aruncării oblice ı̂n cazul când proiectilul ı̂ntâmpină o
fort, ă de rezistent, ă proport, ională cu viteza:

m
dv

dt
= −mgj − bv, (5)

unde b e o constantă care depinde de forma obiectului s, i de proprietăt, ile mediu-
lui. Ecuat, ia are solut, ie analitică:

v = v0e
−bt/m − mg

b

(
1− e−bt/m

)
j. (6)

În continuare vom folosi Maxima pentru a obt, ine această solut, ie.
Pentru ı̂nceput, să definim variabilele dvx s, i dvy care exprimă ecuat, iile

diferent, iale pe axele x s, i y aferente ec. (5):

dvx:m*’diff(vx(t),t)=-b*vx(t)$

dvy:m*’diff(vy(t),t)=-m*g-b*vy(t)$

Atent, ie! Uneori apostroful care precede comanda diff nu este copiat cu co-
manda copy. Este vorba despre caracterul ’, care se găses,te pe tastatură lângă
tasta Enter.

Să observăm că dependent,a funct, iilor vx s, i vy de variabila t este indicată ex-
plicit. Pentru a evita evaluarea prematură a operatorului diferent, ial, acesta este
precedat de o ghilimea dreaptă. Suplimentăm ecuat, iile de mai sus introducând
condit, iile init, iale v(t = 0) = v0:

atvalue(vx(t),t=0,vx0)$

atvalue(vy(t),t=0,vy0)$

Comanda care rezolvă ecuat, iile diferent, iale dvx s, i dvy este desolve:

radcan(desolve([dvx,dvy],[vx(t),vy(t)]));

Pe prima pozit, ie apar ecuat, iile ı̂ntr-o listă, iar pe a doua pozit, ie apar funct, iile
care trebuie determinate.

Problema 8. Pentru cazul aruncării oblice ı̂n mediu cu rezistivitate liniară,
să se determine:

1. Valoarea asimptotică a vitezei când t→∞, folosind comanda limit.

2. x(t), rezolvând ecuat, ia diferent, ială dx
dt = v ı̂mpreună cu condit, iile init, iale

x(t = 0) = 0.

Pentru a compara evolut, ia vitezei ı̂n timp fat, ă de cazul fără rezistent, ă, mai
ı̂ntâi definim variabilele care t, in vitezele din cazul curent:

vlinx:rhs(%o??[1])$

vliny:rhs(%o??[2])$

Semnele de ı̂ntrebare se referă la numărul de ordine al rezultatului rulării co-
menzii desolve.

block([m:1,b:0.05,vx0:9],

plot2d([vlinx,vx0],[t,0,100])

);

block([m:1,b:0.05,vy0:12,g:9.8],

plot2d([vliny,vy0-g*t],[t,0,50])

);

Rezultatele comenzilor de mai sus pot fi vizualizate ı̂n fig. 4

4 Aruncarea oblică ı̂n mediu cu rezistivitate pă-
tratică

Să considerăm că fort,a de rezistent, ă depinde de pătratul vitezei particulei:

m
dv

dt
= −mgj − cvv. (7)

Figura 5: Graficul traiectoriei (y ı̂n funct, ie de x) pentru aruncarea oblică ı̂n
cazul fără rezistent, ă, cu rezistent, ă liniară (b = 10−6) s, i cu rezistent, ă pătratică
(c = 10−6). Viteza init, ială are modulul v0 = 15, unghiul de proiectare este de
45◦ iar masa este 4, 2× 10−6.

Deoarece nu se cunoas,te solut, ia analitică a acestei ecuat, ii, suntem nevoit, i să
recurgem la metode numerice. În primul rând, trebuie definite mărimile care
intervin ı̂n problemă:

v0:15$ deg:%pi/180.0$ theta:45.0*deg$

m:4.2e-6$ c:1e-6$ g:9.8$

v0x:float(v0*cos(theta))$ v0y:float(v0*sin(theta))$

Modulul vitezei la t = 0 este v0 = 15, unghiul de aruncare este de 45◦, masa
particulei este de 4, 2× 10−6 iar coeficientul de rezistent, ă este c = 10−6. Varia-
bila deg are valoarea π/180◦, fiind utilă pentru conversia ı̂ntre grade s, i radiani.
Solut, ia numerică se poate obt, ine folosind algoritmul Runge-Kutta de ordinul 4,
pe care ı̂l apelăm cu funct, ia rk:

tmax:2$

data:rk([vx,vy,-c*sqrt(vx^2+vy^2)*vx/m,-g-c*sqrt(vx^2+vy^2)*vy/m],

[x,y,vx,vy],[0,0,v0x,v0y],[t,0,tmax,tmax/100])$

vals:makelist([data[i][2],data[i][3]],i,1,length(data))$

În comanda rk, pe prima pozit, ie este specificată o listă de ecuat, ii care vor fi
rezolvate numeric simultan. Pe pozit, ia a doua apare lista de variabile (x, y, vx s, i
vy), iar pe pozit, ia a treia apare lista cu valorile init, iale pentru aceste variabile.
Pe ultima pozit, ie se specifică variabila de integrare (t) ı̂mpreună cu domeniul
acesteia (̂ıntre 0 s, i tmax), respectiv pasul temporal sau distant,a ı̂ntre două valori
succesive ale variabilei de integrare (̂ın cazul nostru, folosim 100 de puncte, deci
pasul de timp este tmax/100). Pentru compararea traiectoriei cu rezultatele de
mai ı̂nainte, putem rula comenzile:

xliber:v0x*t$

yliber:v0y*t-g*t^2/2$

b:1e-6$

xlin:(%e^(-(b*t)/m)*(m*%e^((b*t)/m)-m)*v0x)/b$

ylin:(%e^(-(b*t)/m)*(-g*m^2+(g*m^2-b*g*m*t)*%e^((b*t)/m)+

(b*m*%e^((b*t)/m)-b*m)*v0y))/b^2$

plot2d([[parametric,xliber,yliber,[t,0,tmax]],

[parametric,xlin,ylin,[t,0,tmax]],

[discrete,vals]]);

Rezultatul poate fi vizualizat ı̂n Fig. 5.

